

[Description](#)  
[Options](#)[Quick start](#)  
[Remarks and examples](#)[Menu](#)  
[Reference](#)[Syntax](#)  
[Also see](#)

## Description

A paired-coordinate spike plot draws a spike (or line) for each observation in the dataset. The line starts at the coordinate ( $y1var, x1var$ ) and ends at the coordinate ( $y2var, x2var$ ).

## Quick start

Paired-coordinate spike plot with spikes from ( $y1, x1$ ) to ( $y2, x2$ )

```
twoway pcspike y1 x1 y2 x2
```

Same as above, with red lines

```
twoway pcspike y1 x1 y2 x2, lcolor(red)
```

## Menu

Graphics > Two-way graph (scatter, line, etc.)

# Syntax

```
twoway pcspike y1var x1var y2var x2var [if] [in] [, options]
```

<i>options</i>	Description
<code>vertical</code>	orient plot naturally; the default
<code>horizontal</code>	orient plot transposing <i>y</i> and <i>x</i> values
<code>line_options</code>	change look of spike lines
<code>colorvar_options</code>	change color of spike lines based on values of a variable
<code>axis_choice_options</code>	associate plot with alternative axis
<code>twoway_options</code>	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

All explicit options are *rightmost*, except `vertical` and `horizontal`, which are *unique*; see [G-4] **Concept: repeated options**.

# Options

- `vertical` and `horizontal` specify whether the *y* and *x* coordinates are to be swapped before plotting—`vertical` (the default) does not swap the coordinates, whereas `horizontal` does.
- These options are rarely used when plotting only paired-coordinate data; they can, however, be used to good effect when combining paired-coordinate plots with range plots, such as `twoway rspike` or `twoway rbar`; see [G-2] **graph twoway rspike** and [G-2] **graph twoway rbar**.
- `line_options` specify the look of the lines used to draw the spikes, including pattern, width, and color; see [G-3] **line\_options**.
- `colorvar_options` specify that the color of the lines used to draw the spikes be determined by the levels of the numeric variable `colorvar`; see [G-3] **colorvar\_options**.
- `axis_choice_options` associate the plot with a particular *y* or *x* axis on the graph; see [G-3] **axis\_choice\_options**.
- `twoway_options` are a set of common options supported by all `twoway` graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over `by()` groups, and change some advanced settings. See [G-3] **twoway\_options**.

# Remarks and examples

Remarks are presented under the following headings:

- Basic use*
- Advanced use*
- Advanced use 2*

Basic use


We have longitudinal data from 1968 and 1988 on the earnings and total experience of US women by occupation.

```
. use https://www.stata-press.com/data/r19/nlswide1
(National Longitudinal Survey of Young Women, 14-24 years old in 1968)
. list occ wage68 ttl_exp68 wage88 ttl_exp88
```

	occ	wage68	ttl_e-68	wage88	ttl_e-88
1.	Professionals	6.121874	.860618	10.94776	14.11177
2.	Managers	5.426208	1.354167	11.53928	13.88886
3.	Sales	4.836701	.9896552	7.290306	12.62823
4.	Clerical/Unskilled	4.088309	.640812	9.612672	11.08019
5.	Craftsmen	4.721373	1.091346	7.839769	12.64364
6.	Operatives	4.364782	.7959284	5.893025	11.99362
7.	Transport	1.987857	.5247414	3.200494	8.710394
8.	Laborers	3.724821	.775966	5.264415	10.56182
9.	13	5.58524	.8278245	8.628641	12.78389

We graph a spike showing the movement from 1968 values to 1988 values for each observation (each occupation):


```
. twoway pcspike wage68 ttl_exp68 wage88 ttl_exp88
```



## Advanced use

twoway pcspike can be usefully combined with other twoway plottypes (see [G-2] [graph twoway](#)). Here we add markers and labeled markers along with titles and such to improve the graph:


```
. twoway pcspike wage68 ttl_exp68 wage88 ttl_exp88      ||
  scatter wage68 ttl_exp68, msym(0)                  ||
  scatter wage88 ttl_exp88, msym(0) pstyle(p4)
  mlabel(occ) xscale(range(17))
  title("Change in US women's experience and earnings")
  subtitle("by occupation, 1968 to 1988")
  ytitle(Earnings) xtitle(Total experience)
  note("Source: National Longitudinal Survey of Young Women")
  legend(order(2 "1968" 3 "1988"))
```



## Advanced use 2

Drawing the edges of network diagrams is often easier with `twoway pcspike` than with other plot-types.


```
. use https://www.stata-press.com/data/r19/network1  
(Fictional network diagram data)  
. twoway pcspike y_c x_c y_l x_l
```



As with our first example, this graph can be made prettier by combining `twoway pcspike` with other plottypes.

```
. use https://www.stata-press.com/data/r19/network1a
(Fictional network diagram data)

. twoway pcspike y_c x_c y_l x_l, pstyle(p3) ||
    pcspike y_c x_c y_r x_r, pstyle(p4) ||
    scatter y_l x_l, pstyle(p3) msize(vlarge) msym(0)
        mlabel(lab_l) mlabpos(9) ||
    scatter y_c x_c, pstyle(p5) msize(vlarge) msym(0) ||
    scatter y_r x_r, pstyle(p4) msize(vlarge) msym(0)
        mlabel(lab_r) mlabpos(3)
    yscale(off) xscale(off) ylabel(, nogrid) legend(off)
    plotregion(margin(30 15 3 3))
```



## Reference

Cox, N. J. 2009. [Speaking Stata: Paired, parallel, or profile plots for changes, correlations, and other comparisons](#). *Stata Journal* 9: 621–639.

## Also see

[G-2] [graph twoway](#) — Two-way graphs

[G-2] [graph twoway line](#) — Two-way line plots

[G-2] [graph twoway pcarrow](#) — Paired-coordinate plot with arrows

[G-2] [graph twoway pccapsym](#) — Paired-coordinate plot with spikes and marker symbols

[G-2] [graph twoway pci](#) — Two-way paired-coordinate plot with immediate arguments

[G-2] [graph twoway pcscatter](#) — Paired-coordinate plot with markers

[G-2] [graph twoway rspike](#) — Range plot with spikes

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

